


Inventor's Kit Experiment 1 - Say Hello to the BBC micro:bit

Two buttons on the breakout board duplicate the action of the onboard buttons A and B. The program creates displays on the LEDs when the buttons are pressed: One button shows a smiley face image, the other shows the message "Hello".

In the Mr Bit control system, one control module creates the image, whilst the other creates the message.




Here is the program script for the modules:

```
Program Script
module 1
When button A is pressed, show the LED image (Smiley) until button A is free.
module 2
When button B is pressed, show the LED message "Hello" until button B is free.
```

Experiment 2 - Using an LDR & Analogue Input

An LDR (Light Dependent Resistor) is connected in series with a resistor to make a potential divider which creates a voltage that varies with the amount of light shining on the LDR, thus indicating 'light level'. Connecting the voltage to one of the input pins allows the micro:bit to control the LED display to show a different image for bright and dark.

In the Mr Bit control system, one module displays the 'light' image (Sun), whilst the second displays the 'dark' image (Moon).




```
Program Script
module 1
When it is light, show the LED image (Sun) until it is dark.
module 2
When it is dark, show the LED image (Moon) until it is light.
```

Inventors Kit Experiment 3 - Dimming an LED using a Potentiometer

A potentiometer provides an input voltage which can be varied as you turn the control spindle. This voltage is used to alter the brightness of an LED when it is turned on by pressing button A. The higher the voltage from the potentiometer the brighter the LED shines.

The Mr Bit control system contains just one module which uses the potentiometer voltage to control the brightness of the LED using the micro:bit's pulse code modulation function.




```
Program Script
module
When button A is pressed, switch on the LED (potentiometer) until button A is free.
```

Experiment 4 - Using a Transistor to Drive a Motor

The amount of current available from the output pins of the micro:bit is nowhere near large enough to activate a small electric motor. The problem is solved by connecting a transistor to the output and using the tiny current to the transistor to control a much bigger current through the motor.

In the Mr Bit control system the speed of the motor is controlled by the variable *speed* which varies up and down between 0 and 100%. Module 2 counts at one second intervals up to 10 and then back down to 0. The value of the *counter* is scaled up by a factor of 10 by Module 3.




```
Program Script
module 1
Switch on the motor (forwards) (speed) until exit.
module 2
Count every second until the counter is greater than 10.
Count (down) every second until the counter is less than 1.
module 3
Calculate speed = counter x 10 until exit.
```

Inventors Kit Experiment 5 - Using the Accelerometer to Control Motor Speed

The BBC micro:bit has a built in accelerometer which detects the motion of the micro:bit or the strength of gravity in three directions. In this example the accelerometer data is used to measure, starting from a horizontal plane, the tilt angle as the micro:bit is tilted forward and backwards. Taking the analogy of aeroplane motion, this is called the *pitch* angle.

In the Mr Bit control system, the module increases the speed of the motor according to the value of the pitch angle, so the speed varies as you tilt the micro:bit.




```
Program Script
module
Switch on the motor (forwards) (pitch) until exit.
```

Inventors Kit Experiment 6 - Setting the Tone with a Piezo Sounder

The Piezo sounder makes a sound when it is connected to a rapidly alternating voltage (between 0V and 3V). The frequency of the alternating voltage determines the pitch of the tone; pitch increases with frequency.

In the Mr Bit control system module 1 produces a single tone when button A gets pressed; module 2 produces a sequence of tones when button B gets pressed.




```
Program Script
module 1
When button A gets pressed, switch on the piezo sounder for 0.5 seconds.
module 2
When button B gets pressed, switch on the piezo sounder for 5 seconds.
```

Experiment 7 - Wind Power

Wind turbines contain a dynamo which converts the kinetic energy of air into electrical energy. In this experiment the motor is used in reverse like a dynamo to generate a voltage which can be measured and displayed by the BBC micro:bit.

In the Mr Bit control system, module 1 records the *highest voltage* produced as the motor turns. Module 2 displays this on the LEDs and module 3 resets the stored value to zero.




```

Program Script
module 1
  When the voltage is greater than (highest voltage), set highest voltage to voltage.
module 2
  Show the LED message (highest voltage) for 2 seconds. Repeat.
module 3
  When button A gets pressed, set highest voltage to 0.
  
```

Inventors Kit Experiment 8 - Making a Game using the Compass

The BBC micro:bit comes with a built in compass so it can tell which way is north. It does this using a chip that is very sensitive to magnetism and can detect earth's magnetic field. In this game a secret compass heading is selected randomly and an LED blinks faster or slower depending on how close to the correct heading the BBC micro:bit is facing.



In the Mr Bit control system, module 1 creates *target*, the random heading. Module 2 calculates the *difference* between this and the current *compass* reading. Module 3 gives the *result* always as a positive value. Modules 4 and 5 make the LED flash when the result is not small. If the *result* is very small showing that the compass reading is close to the *target*, module 6 switches on the LED continuously.

```


Program Script
module 1
  When button A gets pressed, set target to a random number from 0 to 360.
module 2
  Calculate difference = target - compass until exit.
module 3
  Calculate result = ABS(difference) until exit.
module 4
  When the result is greater than 100, flash the LED until the result is less than 100.
module 5
  When the result is less than 100 and the result is greater than 20, flash the LED until
  the result is less than 20 or the result is greater than 100.
module 6
  When the result is less than 20, switch on the LED until the result is greater than 20.

```

Inventors Kit Experiment 9 - Capacitor Charge Circuit

This experiment uses the analogue input of the BBC micro:bit to measure the voltage at the positive leg of a capacitor. The capacitor circuit allows the capacitor to be charged through a variable resistor. As the capacitor charges, the micro:bit LEDs show the voltage whilst the external coloured LEDs light up according to the magnitude of the voltage.

In the Mr Bit control system, module 1 shows the voltage as a percentage of maximum (3V) and the other modules use the voltage value to switch on the external LEDs.



```


Program Script
module 1
  Show the LED message (voltage) until exit.
module 2
  When the voltage is greater than 25, switch on the Red until the voltage is less than 25.
module 3
  When the voltage is greater than 50, switch on the Orange until the voltage is less than 50.
module 4
  When the voltage is greater than 75, switch on the Yellow until the voltage is less than 75.
module 5
  When the voltage is greater than 95, switch on the Green until the voltage is less than 95.

```

Inventors Kit Experiment 10 - Using an RGB LED

An RGB LED is a special LED that contains three separate LEDs (Red, Green and Blue) in one capsule. By controlling the brightness of these LEDs, their light mixes together to create many different colours.

In the Mr Bit control system, modules 1-3 switch on the three colours, modules 4 – 6 increase the brightness level each time a button is pressed, and modules 7-9 reset the brightness from maximum to zero.



```
Program Script
module 1
  Switch on the Red (red level) until exit.
module 2
  Switch on the Blue (blue level) until exit.
module 3
  Switch on the Green (green level) until exit.
module 4
  When the button at 2 gets pressed, add 10 to red level.
module 5
  When the button at 1 gets pressed, add 10 to blue level.
module 6
  When the button at 0 gets pressed, add 10 to green level.
module 7
  When the red level is greater than 100, set red level to 0.
module 8
  When the blue level is greater than 100, set blue level to 0.
module 9
  When the green level is greater than 100, set green level to 0.
```